// SPDX-License-Identifier: MIT
pragma solidity ^0.6.12;
pragma experimental ABIEncoderV2;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/token/ERC20/SafeERC20.sol";
import "@openzeppelin/contracts-upgradeable/access/OwnableUpgradeable.sol";
import "@openzeppelin/contracts-upgradeable/math/SafeMathUpgradeable.sol";
import "@openzeppelin/contracts-upgradeable/utils/ReentrancyGuardUpgradeable.sol";

import "./interfaces/badger/ISwapStrategyRouter.sol";
import "./interfaces/bridge/IGateway.sol";
import "./interfaces/bridge/IBridgeVault.sol";
import "./interfaces/bridge/ICurveTokenWrapper.sol";
import "./interfaces/curve/ICurveFi.sol";
import "./interfaces/defidollar/IBadgerSettPeak.sol";
import "./interfaces/defidollar/IBadgerYearnWbtcPeak.sol";

contract BadgerBridgeAdapter is OwnableUpgradeable, ReentrancyGuardUpgradeable {
  using SafeMathUpgradeable for uint256;
  using SafeERC20 for IERC20;

  IERC20 public renBTC;
  IERC20 public wBTC;

  // RenVM gateway registry.
  IGatewayRegistry public registry;
  // Swap router that handles swap routing optimizations.
  ISwapStrategyRouter public router;

  event RecoverStuck(uint256 amount, uint256 fee);
  event Mint(uint256 renbtc_minted, uint256 wbtc_swapped, uint256 fee);
  event Burn(uint256 renbtc_burned, uint256 wbtc_transferred, uint256 fee);
  event SwapError(bytes error);

  address public rewards;
  address public governance;

  uint256 public mintFeeBps;
  uint256 public burnFeeBps;
  uint256 private percentageFeeRewardsBps;
  uint256 private percentageFeeGovernanceBps;

  uint256 public constant MAX_BPS = 10000;

  mapping(address => bool) public approvedVaults;

  // Configurable permissionless curve lp token wrapper.
  address curveTokenWrapper;

  // Make struct for mint args, otherwise too many local vars (stack too deep).
  struct MintArguments {
    uint256 _mintAmount;
    uint256 _mintAmountMinusFee;
    uint256 _fee;
    uint256 _slippage;
    address _vault;
    address _user;
    address _token;
    bool _mintIbbtc;
  }

  IERC20 public ibBTC;
  // Peak contract for minting ibbtc from renbtc, sbtc, tbtc.
  IBadgerSettPeak public settPeak;
  // Peak contract for minting ibbtc from wbtc.
  IBadgerYearnWbtcPeak public yearnWbtcPeak;

  mapping(address => uint256) public vaultToPoolId;

  function initialize(
    address _governance,
    address _rewards,
    address _registry,
    address _router,
    address _wbtc,
    uint256[4] memory _feeConfig
  ) public initializer {
    __Ownable_init();
    __ReentrancyGuard_init();

    require(_governance != address(0x0), "must set governance address");
    require(_rewards != address(0x0), "must set rewards address");
    require(_registry != address(0x0), "must set registry address");
    require(_router != address(0x0), "must set router address");
    require(_wbtc != address(0x0), "must set wBTC address");

    governance = _governance;
    rewards = _rewards;

    registry = IGatewayRegistry(_registry);
    router = ISwapStrategyRouter(_router);
    renBTC = registry.getTokenBySymbol("BTC");
    wBTC = IERC20(_wbtc);

    mintFeeBps = _feeConfig[0];
    burnFeeBps = _feeConfig[1];
    percentageFeeRewardsBps = _feeConfig[2];
    percentageFeeGovernanceBps = _feeConfig[3];
  }

  function version() external pure returns (string memory) {
    return "1.1";
  }

  // NB: This recovery fn only works for the BTC gateway (hardcoded and only one supported in this adapter).
  function recoverStuck(
    // encoded user args
    bytes calldata encoded,
    // darkdnode args
    uint256 _amount,
    bytes32 _nHash,
    bytes calldata _sig
  ) external nonReentrant {
    // Ensure sender matches sender of original tx.
    uint256 start = encoded.length - 32;
    address sender = abi.decode(encoded[start:], (address));
    require(sender == msg.sender);

    bytes32 pHash = keccak256(encoded);
    uint256 _mintAmount = registry.getGatewayBySymbol("BTC").mint(
      pHash,
      _amount,
      _nHash,
      _sig
    );
    uint256 _fee = _processFee(renBTC, _mintAmount, mintFeeBps);

    emit RecoverStuck(_mintAmount, _fee);

    renBTC.safeTransfer(msg.sender, _mintAmount.sub(_fee));
  }

  function mint(
    // user args
    address _token, // either renBTC or wBTC
    uint256 _slippage,
    address _user,
    address _vault,
    bool _mintIbbtc,
    // darknode args
    uint256 _amount,
    bytes32 _nHash,
    bytes calldata _sig
  ) external nonReentrant {
    require(
      _token == address(renBTC) || _token == address(wBTC),
      "invalid token address"
    );

    // Mint renBTC tokens
    bytes32 pHash = keccak256(abi.encode(_token, _slippage, _user, _vault));
    uint256 mintAmount = registry.getGatewayBySymbol("BTC").mint(
      pHash,
      _amount,
      _nHash,
      _sig
    );

    require(mintAmount > 0, "zero mint amount");

    uint256 fee = _processFee(renBTC, mintAmount, mintFeeBps);
    uint256 mintAmountMinusFee = mintAmount.sub(fee);

    MintArguments memory args = MintArguments(
      mintAmount,
      mintAmountMinusFee,
      fee,
      _slippage,
      _vault,
      _user,
      _token,
      _mintIbbtc
    );
    bool success = mintAdapter(args);

    if (!success) {
      renBTC.safeTransfer(_user, mintAmountMinusFee);
    }
  }

  function burn(
    // user args
    address _token, // either renBTC or wBTC
    address _vault,
    uint256 _slippage,
    bytes calldata _btcDestination,
    uint256 _amount,
    bool _burnIbbtc
  ) external nonReentrant {
    require(
      _token == address(renBTC) || _token == address(wBTC),
      "invalid token address"
    );
    require(
      !(_vault != address(0) && !approvedVaults[_vault]),
      "Vault not approved"
    );

    bool isVault = _vault != address(0);
    bool isRenBTC = _token == address(renBTC);
    IERC20 token = isRenBTC ? renBTC : wBTC;
    uint256 startBalanceRenBTC = renBTC.balanceOf(address(this));
    uint256 startBalanceWBTC = wBTC.balanceOf(address(this));

    if (_burnIbbtc) {
      //redeem ibbtc
      ibBTC.safeTransferFrom(msg.sender, address(this), _amount);
      uint256 poolId = vaultToPoolId[_vault];
      uint256 redeemAmount = 0;
      if (poolId == 3) {
        redeemAmount = yearnWbtcPeak.redeem(_amount);
      } else {
        redeemAmount = settPeak.redeem(poolId, _amount);
      }
    }

    // Vaults can require up to two levels of unwrapping.
    if (isVault) {
      // First level of unwrapping for sett tokens.
      if (!_burnIbbtc) {
        IERC20(_vault).safeTransferFrom(msg.sender, address(this), _amount);
      }
      IERC20 vaultToken = IBridgeVault(_vault).token();

      uint256 beforeBalance = vaultToken.balanceOf(address(this));
      IBridgeVault(_vault).withdraw(IERC20(_vault).balanceOf(address(this)));
      uint256 balance = vaultToken.balanceOf(address(this)).sub(beforeBalance);

      // If the vault token does not match requested burn token, then we need to further unwrap
      // vault token (e.g. withdrawing from crv sett gets us crv lp tokens which need to be unwrapped to renbtc).
      if (address(vaultToken) != _token) {
        vaultToken.safeTransfer(curveTokenWrapper, balance);
        ICurveTokenWrapper(curveTokenWrapper).unwrap(_vault);
      }
    } else {
      token.safeTransferFrom(msg.sender, address(this), _amount);
    }

    uint256 wbtcTransferred = wBTC.balanceOf(address(this)).sub(
      startBalanceWBTC
    );

    if (!isRenBTC) {
      _swapWBTCForRenBTC(wbtcTransferred, _slippage);
    }

    uint256 toBurnAmount = renBTC.balanceOf(address(this)).sub(
      startBalanceRenBTC
    );
    uint256 fee = _processFee(renBTC, toBurnAmount, burnFeeBps);

    uint256 burnAmount = registry.getGatewayBySymbol("BTC").burn(
      _btcDestination,
      toBurnAmount.sub(fee)
    );

    emit Burn(burnAmount, wbtcTransferred, fee);
  }

  function mintAdapter(MintArguments memory args) internal returns (bool) {
    if (args._vault != address(0) && !approvedVaults[args._vault]) {
      return false;
    }

    uint256 wbtcExchanged;
    bool isVault = args._vault != address(0);
    bool isRenBTC = args._token == address(renBTC);
    IERC20 token = isRenBTC ? renBTC : wBTC;

    if (!isRenBTC) {
      // Try and swap and transfer wbtc if token wbtc specified.
      uint256 startBalance = token.balanceOf(address(this));
      if (!_swapRenBTCForWBTC(args._mintAmountMinusFee, args._slippage)) {
        return false;
      }
      uint256 endBalance = token.balanceOf(address(this));
      wbtcExchanged = endBalance.sub(startBalance);
    }

    emit Mint(args._mintAmount, wbtcExchanged, args._fee);

    uint256 amount = isRenBTC ? args._mintAmountMinusFee : wbtcExchanged;

    if (!isVault) {
      token.safeTransfer(args._user, amount);
      return true;
    }

    // If the token is wBTC then we just approve spend and deposit directly into the wbtc vault.
    if (args._token == address(wBTC)) {
      token.safeApprove(args._vault, token.balanceOf(address(this)));
    } else {
      // Otherwise, we need to wrap the token before depositing into vault.
      // We currently only support wrapping renbtc into curve lp tokens.
      // NB: The curve token wrapper contract is permissionless, we must transfer renbtc over
      // and it will transfer back wrapped lp tokens.
      token.safeTransfer(curveTokenWrapper, amount);
      amount = ICurveTokenWrapper(curveTokenWrapper).wrap(args._vault);
      IBridgeVault(args._vault).token().safeApprove(args._vault, amount);
    }

    if (args._mintIbbtc) {
      IBridgeVault(args._vault).depositFor(address(this), amount);
      uint256 balance = IERC20(args._vault).balanceOf(address(this));

      _mintIbbtc(args._vault, args._user, balance);
    } else {
      IBridgeVault(args._vault).depositFor(args._user, amount);
    }

    return true;
  }

  function _mintIbbtc(
    address _vault,
    address _user,
    uint256 _amount
  ) internal {
    uint256 poolId = vaultToPoolId[_vault];
    uint256 mintAmount = 0;
    //passing in empty variable, no longer used in peak contract
    bytes32[] memory unused;
    if (poolId == 3) {
      IERC20(_vault).approve(address(yearnWbtcPeak), _amount);
      mintAmount = yearnWbtcPeak.mint(_amount, unused);
    } else {
      IERC20(_vault).approve(address(settPeak), _amount);
      mintAmount = settPeak.mint(poolId, _amount, unused);
    }
    ibBTC.safeTransfer(_user, mintAmount);
  }

  function _swapWBTCForRenBTC(uint256 _amount, uint256 _slippage) internal {
    (address strategy, uint256 estimatedAmount) = router.optimizeSwap(
      address(wBTC),
      address(renBTC),
      _amount
    );
    uint256 minAmount = _minAmount(_slippage, _amount);
    require(estimatedAmount > minAmount, "slippage too high");

    // Approve strategy for spending of wbtc.
    wBTC.safeApprove(strategy, _amount);
    uint256 amount = ISwapStrategy(strategy).swapTokens(
      address(wBTC),
      address(renBTC),
      _amount,
      _slippage
    );
    require(amount > minAmount, "swapped amount less than min amount");
  }

  // Avoid reverting on mint (renBTC -> wBTC swap) since we cannot roll back that transaction.:
  function _swapRenBTCForWBTC(uint256 _amount, uint256 _slippage)
    internal
    returns (bool)
  {
    (address strategy, uint256 estimatedAmount) = router.optimizeSwap(
      address(renBTC),
      address(wBTC),
      _amount
    );
    uint256 minAmount = _minAmount(_slippage, _amount);
    if (minAmount > estimatedAmount) {
      // Do not swap if slippage is too high;
      return false;
    }

    // Approve strategy for spending of renbtc.
    renBTC.safeApprove(strategy, _amount);
    try
      ISwapStrategy(strategy).swapTokens(
        address(renBTC),
        address(wBTC),
        _amount,
        _slippage
      )
    {
      return true;
    } catch (bytes memory _error) {
      emit SwapError(_error);
      return false;
    }
  }

  // Minimum amount w/ slippage applied.
  function _minAmount(uint256 _slippage, uint256 _amount)
    internal
    pure
    returns (uint256)
  {
    _slippage = uint256(1e4).sub(_slippage);
    return _amount.mul(_slippage).div(1e4);
  }

  function _processFee(
    IERC20 token,
    uint256 amount,
    uint256 feeBps
  ) internal returns (uint256) {
    if (feeBps == 0) {
      return 0;
    }
    uint256 fee = amount.mul(feeBps).div(MAX_BPS);
    uint256 governanceFee = fee.mul(percentageFeeGovernanceBps).div(MAX_BPS);
    uint256 rewardsFee = fee.mul(percentageFeeRewardsBps).div(MAX_BPS);
    IERC20(token).safeTransfer(governance, governanceFee);
    IERC20(token).safeTransfer(rewards, rewardsFee);
    return fee;
  }

  // Admin methods.
  function setMintFeeBps(uint256 _mintFeeBps) external onlyOwner {
    require(_mintFeeBps <= MAX_BPS, "badger-bridge-adapter/excessive-mint-fee");
    mintFeeBps = _mintFeeBps;
  }

  function setBurnFeeBps(uint256 _burnFeeBps) external onlyOwner {
    require(_burnFeeBps <= MAX_BPS, "badger-bridge-adapter/excessive-burn-fee");
    burnFeeBps = _burnFeeBps;
  }

  function setPercentageFeeGovernanceBps(uint256 _percentageFeeGovernanceBps)
    external
    onlyOwner
  {
    require(
      _percentageFeeGovernanceBps + percentageFeeRewardsBps <= MAX_BPS,
      "badger-bridge-adapter/excessive-percentage-fee-governance"
    );
    percentageFeeGovernanceBps = _percentageFeeGovernanceBps;
  }

  function setPercentageFeeRewardsBps(uint256 _percentageFeeRewardsBps)
    external
    onlyOwner
  {
    require(
      _percentageFeeRewardsBps + percentageFeeGovernanceBps <= MAX_BPS,
      "badger-bridge-adapter/excessive-percentage-fee-rewards"
    );
    percentageFeeRewardsBps = _percentageFeeRewardsBps;
  }

  function setRewards(address _rewards) external onlyOwner {
    rewards = _rewards;
  }

  function setRouter(address _router) external onlyOwner {
    router = ISwapStrategyRouter(_router);
  }

  function setRegistry(address _registry) external onlyOwner {
    registry = IGatewayRegistry(_registry);
    renBTC = registry.getTokenBySymbol("BTC");
  }

  function setVaultApproval(address _vault, bool _status) external onlyOwner {
    approvedVaults[_vault] = _status;
  }

  function setVaultPoolId(address _vault, uint256 _poolId) external onlyOwner {
    vaultToPoolId[_vault] = _poolId;
  }

  function setIbbtcContracts(
    address _ibbtc,
    address _settPeak,
    address _yearnWbtcPeak
  ) external onlyOwner {
    ibBTC = IERC20(_ibbtc);
    settPeak = IBadgerSettPeak(_settPeak);
    yearnWbtcPeak = IBadgerYearnWbtcPeak(_yearnWbtcPeak);
  }

  function setCurveTokenWrapper(address _wrapper) external onlyOwner {
    curveTokenWrapper = _wrapper;
  }

  // Sweep all tokens and send to governance.
  function sweep() external {
    require(
      msg.sender == governance && msg.sender == tx.origin,
      "caller must be governance"
    );
    // NB: Sanity check but governance should have been set on init and cannot be modified.
    require(governance != address(0x0), "must set governance address");
    address[] memory sweepableTokens = new address[](2);
    sweepableTokens[0] = address(renBTC);
    sweepableTokens[1] = address(wBTC);

    for (uint256 i = 0; i < 2; i++) {
      IERC20 token = IERC20(sweepableTokens[i]);
      uint256 balance = token.balanceOf(address(this));
      if (balance > 0) {
        token.safeTransfer(governance, balance);
      }
    }
  }
}

// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

import "./IERC20.sol";
import "../../math/SafeMath.sol";
import "../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using SafeMath for uint256;
    using Address for address;

    function safeTransfer(IERC20 token, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(IERC20 token, address spender, uint256 value) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        // solhint-disable-next-line max-line-length
        require((value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 newAllowance = token.allowance(address(this), spender).add(value);
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 newAllowance = token.allowance(address(this), spender).sub(value, "SafeERC20: decreased allowance below zero");
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) { // Return data is optional
            // solhint-disable-next-line max-line-length
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

import "../utils/ContextUpgradeable.sol";
import "../proxy/Initializable.sol";
/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract OwnableUpgradeable is Initializable, ContextUpgradeable {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    function __Ownable_init() internal initializer {
        __Context_init_unchained();
        __Ownable_init_unchained();
    }

    function __Ownable_init_unchained() internal initializer {
        address msgSender = _msgSender();
        _owner = msgSender;
        emit OwnershipTransferred(address(0), msgSender);
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        emit OwnershipTransferred(_owner, address(0));
        _owner = address(0);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        emit OwnershipTransferred(_owner, newOwner);
        _owner = newOwner;
    }
    uint256[49] private __gap;
}

// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMathUpgradeable {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        uint256 c = a + b;
        if (c < a) return (false, 0);
        return (true, c);
    }

    /**
     * @dev Returns the substraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        if (b > a) return (false, 0);
        return (true, a - b);
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) return (true, 0);
        uint256 c = a * b;
        if (c / a != b) return (false, 0);
        return (true, c);
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        if (b == 0) return (false, 0);
        return (true, a / b);
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        if (b == 0) return (false, 0);
        return (true, a % b);
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, "SafeMath: addition overflow");
        return c;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b <= a, "SafeMath: subtraction overflow");
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        if (a == 0) return 0;
        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");
        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b > 0, "SafeMath: division by zero");
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b > 0, "SafeMath: modulo by zero");
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b <= a, errorMessage);
        return a - b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryDiv}.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        return a % b;
    }
}

// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;
import "../proxy/Initializable.sol";

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuardUpgradeable is Initializable {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    function __ReentrancyGuard_init() internal initializer {
        __ReentrancyGuard_init_unchained();
    }

    function __ReentrancyGuard_init_unchained() internal initializer {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and make it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
    uint256[49] private __gap;
}

//  SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

// ISwapStrategyRouter performs optimal routing of swaps.
interface ISwapStrategyRouter {
  // Return the optimal rate and the strategy ID.
  function optimizeSwap(
    address _from,
    address _to,
    uint256 _amount
  ) external returns (address strategy, uint256 amount);
}

// ISwapStrategy enforces a standard API for swaps.
interface ISwapStrategy {
  function swapTokens(
    address _from,
    address _to,
    uint256 _amount,
    // Slippage is in bps.
    uint256 _slippage
  ) external returns (uint256 amount);

  // Estimate swap amount returns the swap rate.
  function estimateSwapAmount(
    address _from,
    address _to,
    uint256 _amount
  ) external returns (uint256 amount);
}

//  SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

interface IGateway {
  function mint(
    bytes32 _pHash,
    uint256 _amount,
    bytes32 _nHash,
    bytes calldata _sig
  ) external returns (uint256);

  function burn(bytes calldata _to, uint256 _amount) external returns (uint256);
}

interface IGatewayRegistry {
  function getGatewayBySymbol(string calldata _tokenSymbol)
    external
    view
    returns (IGateway);

  function getTokenBySymbol(string calldata _tokenSymbol)
    external
    view
    returns (IERC20);
}

// SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

interface IBridgeVault is IERC20 {
  function depositFor(address recipient, uint256 amount) external;

  function withdraw(uint256 shares) external;

  function token() external returns (IERC20);
}

// SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

interface ICurveTokenWrapper {
  function wrap(address vault) external returns (uint256);

  function unwrap(address vault) external;
}

// SPDX-License-Identifier: MIT
pragma solidity ^0.6.12;

interface ICurveFi {
  function get_virtual_price() external returns (uint256 out);

  function add_liquidity(
    // renbtc/tbtc pool
    uint256[2] calldata amounts,
    uint256 min_mint_amount
  ) external;

  function add_liquidity(
    // sBTC pool
    uint256[3] calldata amounts,
    uint256 min_mint_amount
  ) external;

  function add_liquidity(
    // bUSD pool
    uint256[4] calldata amounts,
    uint256 min_mint_amount
  ) external;

  function get_dy(
    int128 i,
    int128 j,
    uint256 dx
  ) external returns (uint256 out);

  function get_dy_underlying(
    int128 i,
    int128 j,
    uint256 dx
  ) external returns (uint256 out);

  function exchange(
    int128 i,
    int128 j,
    uint256 dx,
    uint256 min_dy
  ) external;

  function exchange(
    int128 i,
    int128 j,
    uint256 dx,
    uint256 min_dy,
    uint256 deadline
  ) external;

  function exchange_underlying(
    int128 i,
    int128 j,
    uint256 dx,
    uint256 min_dy
  ) external;

  function exchange_underlying(
    int128 i,
    int128 j,
    uint256 dx,
    uint256 min_dy,
    uint256 deadline
  ) external;

  function remove_liquidity(
    uint256 _amount,
    uint256 deadline,
    uint256[2] calldata min_amounts
  ) external;

  function remove_liquidity_imbalance(
    uint256[2] calldata amounts,
    uint256 deadline
  ) external;

  function remove_liquidity_imbalance(
    uint256[3] calldata amounts,
    uint256 max_burn_amount
  ) external;

  function remove_liquidity(uint256 _amount, uint256[3] calldata amounts)
    external;

  function remove_liquidity_imbalance(
    uint256[4] calldata amounts,
    uint256 max_burn_amount
  ) external;

  function remove_liquidity(uint256 _amount, uint256[4] calldata amounts)
    external;

  function remove_liquidity_one_coin(
    uint256 _token_amount,
    int128 i,
    uint256 _min_amount
  ) external;

  function commit_new_parameters(
    int128 amplification,
    int128 new_fee,
    int128 new_admin_fee
  ) external;

  function apply_new_parameters() external;

  function revert_new_parameters() external;

  function commit_transfer_ownership(address _owner) external;

  function apply_transfer_ownership() external;

  function revert_transfer_ownership() external;

  function withdraw_admin_fees() external;

  function coins(int128 arg0) external returns (address out);

  function underlying_coins(int128 arg0) external returns (address out);

  function balances(int128 arg0) external returns (uint256 out);

  function A() external returns (int128 out);

  function fee() external returns (int128 out);

  function admin_fee() external returns (int128 out);

  function owner() external returns (address out);

  function admin_actions_deadline() external returns (uint256 out);

  function transfer_ownership_deadline() external returns (uint256 out);

  function future_A() external returns (int128 out);

  function future_fee() external returns (int128 out);

  function future_admin_fee() external returns (int128 out);

  function future_owner() external returns (address out);

  function calc_withdraw_one_coin(uint256 _token_amount, int128 _i)
    external
    view
    returns (uint256 out);

  function calc_token_amount(uint256[2] calldata amounts, bool isDeposit)
    external
    view
    returns (uint256);

  function calc_token_amount(uint256[3] calldata amounts, bool isDeposit)
    external
    view
    returns (uint256);

  function calc_token_amount(uint256[4] calldata amounts, bool isDeposit)
    external
    view
    returns (uint256);
}

// SPDX-License-Identifier: MIT

pragma solidity ^0.6.12;

interface IBadgerSettPeak {
  function mint(
    uint256 poolId,
    uint256 inAmount,
    bytes32[] calldata merkleProof
  ) external returns (uint256 outAmount);

  function approveContractAccess(address account) external;

  function owner() external view returns (address _owner);

  function redeem(uint256 poolId, uint256 inAmount)
    external
    returns (uint256 outAmount);

  function calcMint(uint256 poolId, uint256 inAmount)
    external
    view
    returns (uint256 bBTC, uint256 fee);

  function calcRedeem(uint256 poolId, uint256 bBtc)
    external
    view
    returns (
      uint256 sett,
      uint256 fee,
      uint256 max
    );
}

// SPDX-License-Identifier: MIT
pragma solidity ^0.6.12;

interface IBadgerYearnWbtcPeak {
  function mint(uint256 inAmount, bytes32[] calldata merkleProof)
    external
    returns (uint256 outAmount);

  function approveContractAccess(address account) external;

  function owner() external view returns (address _owner);

  function redeem(uint256 inAmount) external returns (uint256 outAmount);

  function calcMint(uint256 inAmount)
    external
    view
    returns (uint256 bBTC, uint256 fee);

  function calcRedeem(uint256 bBtc)
    external
    view
    returns (
      uint256 sett,
      uint256 fee,
      uint256 max
    );
}

// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        uint256 c = a + b;
        if (c < a) return (false, 0);
        return (true, c);
    }

    /**
     * @dev Returns the substraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        if (b > a) return (false, 0);
        return (true, a - b);
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) return (true, 0);
        uint256 c = a * b;
        if (c / a != b) return (false, 0);
        return (true, c);
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        if (b == 0) return (false, 0);
        return (true, a / b);
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        if (b == 0) return (false, 0);
        return (true, a % b);
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, "SafeMath: addition overflow");
        return c;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b <= a, "SafeMath: subtraction overflow");
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        if (a == 0) return 0;
        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");
        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b > 0, "SafeMath: division by zero");
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b > 0, "SafeMath: modulo by zero");
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b <= a, errorMessage);
        return a - b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryDiv}.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        return a % b;
    }
}

// SPDX-License-Identifier: MIT

pragma solidity >=0.6.2 <0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        // solhint-disable-next-line no-inline-assembly
        assembly { size := extcodesize(account) }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-low-level-calls, avoid-call-value
        (bool success, ) = recipient.call{ value: amount }("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain`call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
      return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value, string memory errorMessage) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.call{ value: value }(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data, string memory errorMessage) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.staticcall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    function _verifyCallResult(bool success, bytes memory returndata, string memory errorMessage) private pure returns(bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                // solhint-disable-next-line no-inline-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;
import "../proxy/Initializable.sol";

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with GSN meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract ContextUpgradeable is Initializable {
    function __Context_init() internal initializer {
        __Context_init_unchained();
    }

    function __Context_init_unchained() internal initializer {
    }
    function _msgSender() internal view virtual returns (address payable) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes memory) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
    uint256[50] private __gap;
}

// SPDX-License-Identifier: MIT

// solhint-disable-next-line compiler-version
pragma solidity >=0.4.24 <0.8.0;

import "../utils/AddressUpgradeable.sol";

/**
 * @dev This is a base contract to aid in writing upgradeable contracts, or any kind of contract that will be deployed
 * behind a proxy. Since a proxied contract can't have a constructor, it's common to move constructor logic to an
 * external initializer function, usually called `initialize`. It then becomes necessary to protect this initializer
 * function so it can only be called once. The {initializer} modifier provided by this contract will have this effect.
 *
 * TIP: To avoid leaving the proxy in an uninitialized state, the initializer function should be called as early as
 * possible by providing the encoded function call as the `_data` argument to {UpgradeableProxy-constructor}.
 *
 * CAUTION: When used with inheritance, manual care must be taken to not invoke a parent initializer twice, or to ensure
 * that all initializers are idempotent. This is not verified automatically as constructors are by Solidity.
 */
abstract contract Initializable {

    /**
     * @dev Indicates that the contract has been initialized.
     */
    bool private _initialized;

    /**
     * @dev Indicates that the contract is in the process of being initialized.
     */
    bool private _initializing;

    /**
     * @dev Modifier to protect an initializer function from being invoked twice.
     */
    modifier initializer() {
        require(_initializing || _isConstructor() || !_initialized, "Initializable: contract is already initialized");

        bool isTopLevelCall = !_initializing;
        if (isTopLevelCall) {
            _initializing = true;
            _initialized = true;
        }

        _;

        if (isTopLevelCall) {
            _initializing = false;
        }
    }

    /// @dev Returns true if and only if the function is running in the constructor
    function _isConstructor() private view returns (bool) {
        return !AddressUpgradeable.isContract(address(this));
    }
}

// SPDX-License-Identifier: MIT

pragma solidity >=0.6.2 <0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library AddressUpgradeable {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        // solhint-disable-next-line no-inline-assembly
        assembly { size := extcodesize(account) }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-low-level-calls, avoid-call-value
        (bool success, ) = recipient.call{ value: amount }("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain`call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
      return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value, string memory errorMessage) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.call{ value: value }(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data, string memory errorMessage) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.staticcall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    function _verifyCallResult(bool success, bytes memory returndata, string memory errorMessage) private pure returns(bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                // solhint-disable-next-line no-inline-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}
{"success": true, "error": null, "results": {"detectors": [{"check": "unused-return", "impact": "Medium", "confidence": "Medium"}, {"check": "divide-before-multiply", "impact": "Medium", "confidence": "Medium"}, {"check": "incorrect-equality", "impact": "Medium", "confidence": "High"}, {"check": "uninitialized-local", "impact": "Medium", "confidence": "Medium"}]}}